
'////////////////////////////////////////////////////////////////////////////// 

'//                                                                          // 

'//                      Script de gestion des Imprimantes V1.0              // 

'//                             (C)2005-2010 KoXo Dev.                       // 

'//                               http://www.koxo.net                        // 

'//                                                                          // 

'////////////////////////////////////////////////////////////////////////////// 

'//                                                                          // 

'// Execution : Printers.VBS PrintersFile UnconnPrinters ConnectPrinters     // 

'//                         DisplayMessage                                   // 

'//                                                                          // 

'// PrintersFile : Full path to printer's file                               // 

'// UnconnPrinters  : 0 or 1 : First unconn network printers                 // 

'// ConnectPrinters : 0 or 1 : connect or install printers                   // 

'// DisplayMessage  : 0 or 1 : Display message for connections               // 

'//                  (printers can be installed and conneted by GPO).        // 

'//                                                                          // 

'// Ex : Printers.vbs \\srv\netlogon\printersAdmins.txt 0 1 1                // 

'//                                                                          // 

'////////////////////////////////////////////////////////////////////////////// 

' 

' Default params. 

' 

UnConnPrinters=True 

ConnectPrinters=True 

DisplayMessage=False 

MessageTimeOut=3 

PrintersFile="%LOGONSERVER%\NETLOGON\Printers.txt" 

' 

' Command line options 

' 

On Error Resume Next 

Dim  objShell    

Set objShell=WScript.CreateObject("WScript.Shell") 

LogonServer=objShell.ExpandEnvironmentStrings("%LOGONSERVER%") 

Dim objArg 

Set objArgs=WScript.Arguments 

If objArgs.Count>0 Then  

   For i=0 To objArgs.Count-1  

       If i=0 Then  

          PrintersFile=objArgs.Item(i)  

          End If 

       If i=1 Then  

          UnconnPrinters=(objArgs.Item(i)=1)  

          End If 

       If i=2 Then  

          ConnectPrinters=(objArgs.Item(i)=1)  

          End If 

       If i=3 Then  

          DisplayMessage=(objArgs.Item(i)=1)  

          End If 

   Next     

   Else 

   PrintersFile=Replace(PrintersFile,"%LOGONSERVER%",LogonServer) 

   End If 

' 

' Unconn printers if specified 

' 

Dim objNetwork 

Set objNetwork=WScript.CreateObject("WScript.Network") 

Dim objPrinter 

Set objPrinter=objNetwork.EnumPrinterConnections 

If UnconnPrinters Then 

   If objPrinter.Count>0 Then 

      For i=0 To objPrinter.Count-1 Step 2 

          If InStr(objPrinter.Item(i+1),"\\")>0 Then 

             objNetwork.RemovePrinterConnection(objPrinter.Item(i+1)) 



          End If 

      Next     

      End If     

End If 

' 

' Computer Name 

' 

Computer=objShell.ExpandEnvironmentStrings("%CLIENTNAME%") 

TSE=(Computer<>"" And Computer<>"%CLIENTNAME%" And Computer<>"Console") 

If Not TSE Then  

   Computer=objNetwork.ComputerName  

End If 

' 

' Add printers specified in "Printers.txt" file  

' 

Dim objFSO 

Set objFSO=CreateObject("Scripting.FileSystemObject") 

If (objFSO.FileExists(PrintersFile)) Then   

   Set objTextFile=objFSO.OpenTextFile(PrintersFile,1) 

   Do Until objTextFile.AtEndOfStream 

         strLine=objTextFile.ReadLine 

         i=InStr(strLine,"=") 

         If (i>0) Then 

            ' Room 

            strRoom=Mid(strLine,1,i-1) 

            ' Resources 

            strLine=Mid(strLine,i+1,255) 

            arrstrLine=Split(strLine,",") 

            ' Joker 

            i=InStr(strRoom,"*") 

            If i>0 Then  

               RoomPrefix=Mid(strRoom,1,i-1) 

               Else  

               RoomPrefix=strRoom 

               End If 

            If InStr(UCase(Computer),UCase(RoomPrefix))>0 Then 

               If ConnectPrinters Then  

                  ' Message    

                  Message="Les imprimantes suivantes vont être ajoutées :"+vbCrLf 

                  For i=0 To UBound(arrstrLine)  

                      Message=Message+arrstrLine(i)+vbCrLf  

                      Next 

                  If DisplayMessage Then  

                     objShell.Popup Message,MessageTimeOut,"Connexion d'imprimantes",64  

                     End If 

                  ' Connect 

                  For i=0 To UBound(arrstrLine)  

                      AddWindowsPrinterConnection arrstrLine(i)                       

                  Next 

                  End If 

                If DisplayMessage Then  

                   Message="L'imprimante suivante sera définie par défaut : "+vbCrLf 

                   objShell.Popup Message+arrstrLine(0),MessageTimeOut,"Imprimante par 

défaut",64            

                   End If 

                SetDefaultPrinter arrstrLine(0) 

                Exit Do 

                End If 

            End If 

   Loop 

   objTextFile.Close 

End If 

WScript.Quit 

'////////////////////////////////////////////////////////////////////////////// 

'//                                                                          // 

'//                            Sub-Routines                                  // 

'//                                                                          // 



'////////////////////////////////////////////////////////////////////////////// 

'Set the default printer (local or network) 

Sub SetDefaultPrinter(sUNCPrinter) 

Dim sComputer, sShare 

 sShare=Mid(sUNCPrinter,3,255) 

 sComputer=Mid(sShare,1,InStr(sShare,"\")-1) 

 sShare=Mid(sShare,InStr(sShare,"\")+1,255) 

 If (UCase(objNetwork.ComputerName)=UCase(sComputer)) Then 

    Set objWMIService=GetObject("winmgmts:" _ 

        & "{impersonationLevel=impersonate}!\\" & sComputer & "\root\cimv2") 

    Set colInstalledPrinters=objWMIService.ExecQuery _ 

        ("Select * from Win32_Printer Where ShareName='" & sShare & "'") 

    For Each objPrinter In colInstalledPrinters            

         objPrinter.SetDefaultPrinter() 

        Next 

    Else 

     objNetwork.SetDefaultPrinter sUNCPrinter 

    End If 

End Sub 

'Sub AddWindowsPrinterConnection 

Sub AddWindowsPrinterConnection(sUNCprinter) 

Dim sComputer 

 sComputer=Mid(sUNCPrinter,3,255) 

 sComputer=Mid(sComputer,1,InStr(sComputer,"\")-1) 

 If (UCase(objNetwork.ComputerName)<>UCase(sComputer)) Then 

    objNetwork.AddWindowsPrinterConnection sUNCprinter 

    End If 

End Sub 

'////////////////////////////////////////////////////////////////////////////// 

 


